
his newsletter is the home of Foundation SAS and brings 
you information about making the most of SAS 9 and 
Altair  SLC.   This  issue  has  been  delayed,  because  I 

under-estimated  how  much  effort  organising  my  significant 
birthday celebrations would take!. It will, therefore, cover Q3 and 
Q4  of  2024.   This  newsletter  issue  includes  a  macro  that 
everybody  needs  to  know  for  categorical  data  with  a  group 
variable from LeRoy Bessler,  and  Richard  Carson  gives  us  2 
useful tips.  As usual I have publicised some SAS-related events 
for the next few months, and I have also added some more SAS 
formats,  options and functions that  you may or may not  have 
come across  before,  but  you  might  find  interesting,  and  even 
useful.

T

If  you  would  like  to  contribute  an  article,  to  re-visit  and 
improve  an  existing  article,  or  just  discuss  the  possibility  of 
doing  so,  please  feel  free  to  send  an  email  to  me  at 
views-uk@hollandnumerics.org.uk.

Philip R Holland (Newsletter Editor)

Membership of VIEWS is free and you can register for 
notification of all future VIEWS News issues by registering at 

blog.hollandnumerics.org.uk, and then emailing 
views-uk@hollandnumerics.org.uk, or using the blog site 

menu, to request access to the VIEWS UK forum. Members will 
receive an email notification whenever a new issue of the free 

quarterly VIEWS News is published on the web site, please 
inform us when you change your email address, otherwise your 

registration may be cancelled.

Ask The Consultant
This  part  of  VIEWS  News is  where  you  can  get  your 
technical questions answered.  Send your questions to the 
Editor.

The  Most  Communication-Efficient  and 
Communication-Effective  Bar  Chart  for 
Categorical  Data  that  includes  a  Group 
Variable

Q: What  is  the  most  communication-efficient  and 
communication-effective  bar  chart  for  categorical  data  that 
includes a group variable?

A: [SAS 9.4] The commonest,  though not  best,  way to  handle 
such data is with a stacked vertical bar chart.  I have elsewhere 
shown the better alternative which includes an x-axis table.  Find 
it and other alternatives in Chapter 4 of the book mentioned at 
the end of this article.  Last month, I finally developed my best 
solution.

For categorical data, even without a group variable, I have 
always advocated providing response value and percent of grand 
total for each category, and recently also that the grand total be 
provided in the title  or  subtitle.   With a group variable,  more 
pieces  of  data  come  into  the  picture,  potentially  at  least, 
depending  on  what  you  code,  the  subtotals  for  each  category 
value and for each group value.   There are no ODS Graphics 
features (that I have found) to make that easy, but I found a less 
easy way.  Below is the absolute best way to visually package all 
of  that  information.   I  DID  find  a  stacked  vertical  bar  chart 
solution,  not  shown  here,  that  comes  close,  but  close  isn’t 
complete.

What is distinctive about this result is that the legend colour 
swatches are actually superfluous since each bar does have its 
region  identity  supplied  in  its  bar  label.   Here,  the  important 
function of the legend is serving as a table of region subtotals.

However, in a chart with many more bars, the colour-coding 
makes it easy to find all of the bars for any particular region.  For  
example, it is immediately apparent that the largest share of sales 
is for Slippers in the Pacific Region.  A viewer might be curious 
as to how strong were sales of OTHER products in the Pacific 
Region.  One can just look at the other blue bars to discover that 
without having to READ the bar labels.  Though the legend itself 
is  not  the  tool  in  that  hunt,  the  helpful  colour-coding  is  a 
consequence of its presence.

The code for the example includes a macro that I introduced 
and discussed in VIEWS News Issue 76 Second Quarter 2024. 
See  that  issue  for  more  information  about  the  macro  and 
recommendations related to text for graphs.

%MACRO AllGraphTextSetUp(Size, Family = ARIAL,
                         Weight = BOLD);
 %LET FamilyForStyle =
  %SYSFUNC(COMPRESS(&Family., ' '));

 PROC TEMPLATE;
  DEFINE STYLE
    GraphFont&FamilyForStyle.&Size.Pt&Weight.;
   PARENT = STYLES.LISTING;
   CLASS GraphFontS /
    'GRAPHVALUEFONT' = ("&Family.", &Size.PT,
                        &Weight.)
    'GRAPHLABELFONT' = ("&Family.", &Size.PT,
                        &Weight.)
    'GRAPHDATAFONT' = ("&Family.", &Size.PT,
                       &Weight.)
    'GRAPHTITLEFONT' = ("&Family.", &Size.PT,
                       &Weight.)
    'GRAPHFOOTNOTEFONT' = ("&Family.",
                           &Size.PT, &Weight.);
  END;
 RUN;
%MEND AllGraphTextSetUp;

%AllGraphTextSetUp(11);

%LET GrandTotalFormat  = DOLLAR10.;
%LET SubTotalFormat  = DOLLAR8.;
%LET PercentFormat  = 6.2;

PROC SUMMARY DATA=SASHELP.SHOES;
  WHERE Region IN ('Asia' 'Africa' 'Pacific')
    AND Product IN ('Boot' 'Sandal' 'Slipper');
  CLASS Region Product;
  VAR Sales;
  OUTPUT OUT = Totals SUM = Sales;
RUN;

DATA _NULL_;
  SET Totals;
  IF _TYPE_ EQ 0;
  CALL SYMPUT('GrandTotalDisplay',
    TRIM(LEFT(PUT(Sales,&GrandTotalFormat.))));
  CALL SYMPUT('GrandTotal',Sales);
RUN;

DATA RegionAndRegionSales;
  KEEP Region RegionAndRegionSales;
  LENGTH RegionAndRegionSales $64;
  SET Totals;
  WHERE _TYPE_ EQ 2;
  RegionAndRegionSales = TRIM(LEFT(Region))
    || ' '
    || TRIM(LEFT(PUT(Sales,&SubTotalFormat.)))
    || ' '
    || TRIM(LEFT(PUT((Sales / &GrandTotal.) *
                      100,&PercentFormat.)))
    || '%';
RUN;

DATA ProductAndProductSales;
  KEEP Product ProductAndProductSales;
  LENGTH ProductAndProductSales $64;
  SET Totals;
  WHERE _TYPE_ EQ 1;
  ProductAndProductSales = TRIM(LEFT(Product))
    || ' '
    || TRIM(LEFT(PUT(Sales, &SubTotalFormat.)))
    || ' '
    || TRIM(LEFT(PUT((Sales / &GrandTotal.) *
                      100, &PercentFormat.)))
    || '%';
RUN;

/*PREFIX Region-Product Totals WITH Region
  VALUES*/
DATA RegionProductSalesAndPercent;
  KEEP Sales RegionProductSalesAndPercent;
  LENGTH RegionProductSalesAndPercent $64;
  SET Totals;
  WHERE _TYPE_ EQ 3;
  RegionProductSalesAndPercent =
    TRIM(LEFT(Region))
    || ' '
    || TRIM(LEFT(PUT(Sales, &SubTotalFormat.)))
    || ' '
    || TRIM(LEFT(PUT((Sales / &GrandTotal.) *
                      100, &PercentFormat.)))
    || '%';
RUN;

/*Create a format to convert each Region-Product
  Sales value (in each Product Cluster) into a
  string that consists of the Relevant Region
  and that Sales value without Product
  identified in the string*/
DATA CNTLIN;
  LENGTH LABEL $64;
  RETAIN FMTNAME 'ValueAndPercent' TYPE 'N';
  SET RegionProductSalesAndPercent
    (RENAME = (Sales = START
       RegionProductSalesAndPercent = LABEL))
    END = Lastone;
  OUTPUT;
  IF LastOne THEN DO;
    HLO = 'O';
    LABEL = '***ERROR***';
    OUTPUT;
  END;
RUN;

PROC FORMAT LIBRARY = WORK CNTLIN = CNTLIN;
RUN;

DATA RegionProductSales;
  SET Totals;
  WHERE _TYPE_ EQ 3;
RUN;

DATA Nearly;
  DROP Region;
  MERGE RegionProductSales RegionAndRegionSales;
  BY Region;
RUN;

PROC SORT DATA = Nearly;
  BY Product;
RUN;

DATA ToPlot;
  DROP Product;
  MERGE Nearly ProductAndProductSales;
  BY Product;
RUN;

FOOTNOTE; /*erase any leftover*/

ODS LISTING GPATH = "C:\TEMP" DPI = 300
            STYLE = GraphFontArial11ptBold;
ODS GRAPHICS ON / RESET = ALL SCALE = OFF
  WIDTH = 3.65IN HEIGHT = 3.65IN
  IMAGENAME =
   "BestBarChartForCategoriesWithGroupVariable";

TITLE1 J = L
  "Shoe Sales By Region Within Product";
TITLE2 J = L
  "Products Ordered By Product Total Sales";
TITLE3 J = L "Grand Total &GrandTotalDisplay.";

PROC SGPLOT DATA = ToPlot NOBORDER;
  STYLEATTRS DATACOLORS = (blue green purple);
  /*arrange the colour order to your
    preference*/
  HBAR ProductAndProductSales /
    RESPONSE = Sales CATEGORYORDER = RESPDESC
    GROUP = RegionAndRegionSales
    GROUPDISPLAY = CLUSTER DATALABEL
    DATALABELPOS = LEFT BARWIDTH = 0.7
    NOOUTLINE DISPLAYBASELINE = OFF;
  FORMAT Sales ValueAndPercent.;
  YAXIS DISPLAY = (NOLINE NOTICKS NOLABEL)
    SPLITCHAR = ' ' FITPOLICY = SPLITALWAYS;
  XAXIS DISPLAY = NONE;
  LABEL Sales = '00'X
    ProductAndProductSales = '00'X;
  KEYLEGEND / SORTORDER = REVERSEAUTO
    /*match the order of the bars within each
      cluster*/
    TITLE = '' NOBORDER ACROSS = 1
    FILLHEIGHT = 9PT /*match height of labels*/
    FILLASPECT = GOLDEN /*rather than a square,
      use rectangle with more area for colour
      fill*/
    AUTOOUTLINE;
RUN;

ODS LISTING CLOSE;
TITLE;

It is regrettable that this example did not appear in the book 
Visual  Data  Insights  Using  SAS  ODS  Graphics:  A  Guide  to 
Communication-Effective  Data  Visualization.   The  author 
warrants  that  its  contents  are  (in  pharmaceutical  industry 
language) safe and effective.

LeRoy Bessler, Bessler Consulting and Research

Did You Know?
If you have a useful hint or tip, please send it to the Editor, 
and share it with the VIEWS membership.

Convert numeric values to text with no NOTE

[SAS 8 and Altair SLC] Use CATS(number) to convert a value 
in numeric form to text without incurring a NOTE in the log.

CATS(right_aligned_expression)  is  the  equivalent  of 
TRIM(LEFT(right_aligned_expression)), but without the NOTE.

Richard C Carson

PICTURE formats with full month names

[SAS 9 and Altair SLC] To express  a  date  or  datetime as  full 
month name with 4-digit year, e.g. for a report title, create the 
following PICTURE formats:

OPTIONS VALIDFMTNAME = LONG;

PROC FORMAT;
  PICTURE fullmonthyeardt (DEFAULT = 16)
    OTHER = '%B %Y' (DATATYPE = DATE);
  PICTURE fullmonthyeardttm (DEFAULT = 16)
    OTHER = '%B %Y' (DATATYPE = DATETIME);
RUN;

The result will be right-aligned within the default length.

Richard C Carson

Formats, Options, and Functions
This  section  is  devoted  to  the  description  of  useful,  or 
unusual,  SAS Formats,  Options,  and Functions.   Please 
read the SAS documentation to find out more about them.

[SAS] The SAS format  $MSGCASEw. writes character data in 
uppercase when the -MSGCASE system option is in effect.  The 
SAS  configuration  option  -MSGCASE displays  all  warning, 
note and error messages in upper-case.  The SAS configuration 
option  -NOMSGCASE displays  all  warning,  note  and  error 
messages in mixed case, and is the default setting.

[SAS 9 and Altair SLC 2024] PROC JSON can now write 
JSON files using the  PRETTY or  NOPRETTY option.   The 
NOPRETTY option generates smaller JSON files by removing 
all newline characters.

[SAS and Altair SLC 2024] Fuzzy matching of words can be 
done  using  the  SOUNDEX(),  SPEDIS()  and  COMPGED() 
functions.

In Brief
• Back issues of VIEWS News are available from the VIEWS UK 

forum at  blog.hollandnumerics.org.uk.  Please register for 
free membership on the blog, and then request free access to 
the VIEWS UK forum.

• The SAS Professional Forum group on LinkedIn, which 
currently has over 40,900 members.  Everyone is welcome to 
request to join, but discussions here are restricted to 
SAS-related topics: www.linkedin.com/groups?gid=70702.

• The SAS Opportunities group on LinkedIn, which currently 
has over 22,200 members.  Everyone is welcome to request to 
join, but discussions here are also restricted to SAS-related 
topics: www.linkedin.com/groups?gid=805217.

• In the UK and Ireland the SUGUKI group have been holding 
regular meetings in London, online, and also occasional 
meetings in Bristol, Dublin, Edinburgh, Leeds and 
Manchester.  For more details and to register for future events 
please go to meetup.com/SUGUKI.

• Fancy joining the SAS Global Analytics Explorers hub, then 
please fill in your name, company email address and 
company at https://infl.tv/npmv.

• Altair SLC Personal Edition provides named users with free 
access to Altair SLC and Analytics Workbench software 
(formerly WPS) including a SAS language compiler with 
coding and workflow tools on unlimited data volumes, with 
interfaces to a wide range of databases, as well as integrated 
R and Python programming environments.  When the 
Personal Edition software is downloaded, it includes a licence 
for up to 12 months.  At the end of this term, users should 
download a new copy of the software to renew their licence 
and/or to upgrade to the latest available full version.  
Download the software from www.altair.com/altair-slc.  No 
technical support service is offered with the Personal Edition, 
but assistance can be requested using the Community forums 
at community.altair.com/community.  The current General 
Access version of Altair SLC is 5.24.2 (2024).

• Do you use Microsoft Visual Studio Code Editor?

• You may be interested in a SAS-related extension called 
SASjs - DevOps for SAS.  You can develop SAS 
applications and run code on SAS Viya, SAS 9 EBI or 
SASjs Server!  Just set up the target, and click the 
running man to execute your SAS code.  The SASjs 
Server could be your local SAS installation.  SASjs is an 
open source project, and can be obtained from the Visual 
Studio Code Marketplace, where the current version is 
v2.6.0.

• The SAS Institute has also been developing its own 
extension called SAS.  You can develop SAS 
applications and run code on local SAS installations, SAS 
Viya, or SAS OnDemand for Academics!  Just set up the 
target, and click the running man to execute your SAS 
code.  The extension can be obtained from the Visual 
Studio Code Marketplace, where the current version is 
v1.11.0.

• Another Microsoft Visual Studio Code Editor extension has 
just been released called Altair SLC.  This extension 
currently only allows you to connect to local Altair SLC 
installations, even if, until now, you have only been able 
to use Altair Analytics Workbench to access Altair SLC.  
The current version in the Visual Studio Code 
Marketplace is v0.7.1 (pre-release).

Diary
Are you organising an event that would be of interest to the 
VIEWS readership?  Let us know as we are interested in all 
non-profit making events related to SAS.

March 2025
16-19 PhUSE US Connect, Orlando, FL, USA

The  PhUSE  Community  allows  you  to  enhance  your 
learning and gain insights from our experts on today's most 
critical  industry  topics  through  a  variety  of  engaging 
presentations,  workshops,  and  networking  opportunities. 
Individuals from all areas of the industry will be gathered in 
one  place,  providing  an  opportunity  not  only  for  you  to 
advance your career but also to heighten the pharmaceutical 
industry’s  influence  and  findings.   Full  details  at 
www.phuse.eu.

May 2025
6-9 SAS Innovate, Orlando, FL, USA

Whether you're a business leader,  technical user or SAS 
Partner, SAS Innovate is a no-risk, all-reward opportunity to 
learn how to succeed with data and AI.  Full details can be 
found at www.sas.com/en_us/events.

June 2025
1-4 PharmaSUG US, San Diego, CA, USA

Join  us  for  the  premier  event  of  the  year  for  statistical 
programmers,  statisticians,  data  managers,  researchers,  and 
others  who  analyze  data  in  the  life  sciences  -  PharmaSUG 
2025!  Full details at www.pharmasug.org.

Contacts

LeRoy Bessler Email: besslerleroy2024@gmail.com

Richard C Carson Email: richard@arcol-analytics.com

SAS UK Tel: +44-(0)1628-486933
Web: www.sas.com/uk

VIEWS

Web Site Web: blog.hollandnumerics.org.uk

Newsletter Editor Email:
views-uk@hollandnumerics.org.uk

